
In this appendix we provide eight examples from intermediate microeconomic 

and macroeconomic theory.  The first example illustrates utility functions and 

indifference curves.  The second illustrates consumer maximization.  The third represents 

the producer problem of cost minimization.  The last example demonstrates how MAPLE 

can be used to solve a Keynesian macroeconomic system. 

 While students may grasp the concept of an indifference curve, they often have 

difficulty envisioning a single indifference curve as part of an indifference map.  This 

first example can be used to familiarize the student with the basic syntax of MAPLE 

while depicting the indifference curve as a contour line defined by { },x y such that 

( ),f x y k=  where  is a real-valued constant.  Note that in the following examples, 

MAPLE code is preceded by the ">" symbol.  One may include comments by omitting 

this symbol. 

k

Example 1:  
Graph of Cobb-Douglas Utility:  The Indifference Map
> restart;
> u:=3*x^(2/3)*y^(1/3);

> 
plot3d(u,x=0..100,y=0..100,labels=["x","y","utility"],title
="Figure 1: Indifference Map",axes=boxed,thickness=4); 
 

 
 
Graphing Indifference Curves
> y1:=solve(u=100,y); 

> plot(y1,x=0..100,y=0..100,labels=["x","y"],title="Figure 
2: Indifference Curve for 100 
utils",color=black,thickness=4); 

The next example illustrates a solution to the consumer utility maximization 
problem using Lagrange multipliers.  The MAPLE syntax helps one to clearly distinguish 
the felicity (function to be optimized) from the budget constraint.   

Example 2: 
Example of Consumer Optimization with Quasi-linear Preferences
> restart; 
 Type in utility function.
> u:=y+x^(1/2); 

 Type in the price of good x.
> p[x]:=5; 

 Type in the price of good y.
> p[y]:=7; 

 Type in the amount in the budget, M.
> M:=20; 

> Mo:=p[x]*x+p[y]*y-M; 

 Set up the Lagrangian.
> L:=u-lambda*(Mo); 

 Find the first order condition of the Lagrangian with respect to x.
> Diff(L,x); 

> Lx:=value(%); 

 Find the first order condition of the Lagrangian with respect to y.
> Diff(L,y); 

> Ly:=value(%); 

 The following solution (soln) is the optimal consumer bundle.
> soln:=solve({Lx=0,Ly=0,Mo=0},{x,y,lambda}); 

> umx:=eval(u,soln); 

 The following value is the maximum utility level, indicative of successful consumer 
optimization.
> umax:=evalf(umx); 

 Next we specify the budget constraint by defining the budget constraint as a 
function of y in terms of x and income. 
> xint:=M/p[x]; 

> yint:=M/p[y]; 

> BC:=x->yint-(p[x]/p[y])*x; 

> indy:=solve(u=umax,y); 

> indcv:=x->indy; 

 The graph below illustrates the optimal solution to this consumer problem. 
> plot([BC(x),indcv(x)],x=0..xint,labels=["good x","good 
y"],title="Figure 3: Consumer Equilibrium",thickness=3); 
 

 5



 
 
Examine the utility function and the representative indifference curve depicted 
here.   
Can you describe the bias? 
 
Figure 3 clearly illustrates the bias of the quasi-linear utility function towards good y. 

Example three illustrates the similarities and differences between consumer and 
producer theory.  The MAPLE syntax indicates that although they illustrate different 
economic concepts, clearly they are both problems involving constrained maximization.  
The similarities and differences become more evident when the student is requested to 
make certain changes to the template provided by the instructor.  In order to provide more 
of a challenge, the instructor may decide to use the consumer's utility maximization 
problem as the template for the producer's cost minimization problem.  Note the slightly 
more general approach. 

Example 3: 
This program will determine the Minimum Cost with Cobb-
Douglas Production.
> restart; 
> assume(L,nonnegative); 
> assume(K,nonnegative); 
> assume(lambda,nonnegative); 
>  
The Cobb-Douglas Production Function:
> F:=A*L^(delta)*K^(epsilon); 

> "Type in values for A, delta and epsilon."; 

> A:=1; 

> delta:=2/3; 

> epsilon:=1/3; 

Type in the wage.
> w:=2; 

Type in the rental rate.
> r:=5; 

>  
Type in the desired output level.
>  
> q:=100; 

The Cost Function Follows.
> C:=r*K+w*L; 

>  
The output constraint follows: 
Note:  This becomes a minimization problem when Qo<=0.
> Qo:=q-F; 

> Lag:=C+lambda*(Qo); 

> Diff(Lag,L); 

> LL:=value(%); 

> Diff(Lag,K); 

> LK:=value(%); 

The following solution is the cost-minimizing bundle.
>  
> soln:=fsolve({LL=0,LK=0,Qo=0},{L,K,lambda}); 

>  
The following value is the mimimum cost.
> minc:=eval(C,soln); 

> lint:=minc/w; 

> kint:=minc/r; 

> TC:=L->kint-(w/r)*L; 

>  
> isoK:=solve(q=F,K); 

> isoquant:=L->isoK; 

The graph below illustrates the optimal solution.
> 
plot([TC(L),isoquant(L)],L=0..lint,K=0..kint,labels=["Labor
","Capital"],title="Production Equilibrium",thickness=3); 

 

 
To illustrate that this is indeed the minimum cost, we 
evaluate the cost of two nearby points of the same 
isoquant.
> L1:=eval(L,soln)-.2; 

> L2:=eval(L,soln)+.2; 

> K1:=eval(isoquant(L),L=L1); 

> K2:=eval(isoquant(L),L=L2); 

> C1:=eval(w*L1+r*K1); 

> C2:=eval(w*L2+r*K2); 

Recall that the minimum cost was 512.9927841 < C1 and C2.
 

In the next example we solve for and demonstrate the Hicksian substitution and 
income effects. 

Example 4: 
Hicks Substitution and Income Effects
> assume(x,nonnegative);
> assume(y,nonnegative);
> assume(lambda,nonnegative);
 Type in values for A, delta and epsilon.
> A:=1;

> delta:=1/2;

> epsilon:=1/2;

 The Cobb-Douglas Utility Function follows.
> u:=A*x^(delta)*y^(epsilon);

 Type in the price of good x.  We will investigate the effect of a price increase in good 
x.
> p1[x]:=2;

> p2[x]:=3;

 Type in the price of good y.
> p[y]:=50/8;

 Type in the budget amount.
> M:=100;

 Write the budget constraints, before and after the increase in the price of x. 
> M1:=M-(p1[x]*x+p[y]*y);

> M2:=M-(p2[x]*x+p[y]*y);

 Deternine the first Lagrangian equation. 
> L1:=u+lambda[1]*(M1);

> Diff(L1,x);

> L1x:=value(%);

> Diff(L1,y);

> L1y:=value(%);

 Next, determine the Lagrangian equation after the price of x rises. 
> L2:=u+lambda[2]*(M2);

> Diff(L2,x);

> L2x:=value(%);

> Diff(L2,y);

> L2y:=value(%);

 The following solutions (soln1 and soln2) are the optimal bundles before and after 
the rise in the price of x.

 11



> soln1:=solve({L1x=0,L1y=0,M1=0},{x,y,lambda[1]});

> soln2:=solve({L2x=0,L2y=0,M2=0},{x,y,lambda[2]});

> umx1:=eval(u,soln1);

> umx2:=eval(u,soln2);

 The values for umax1 and umax2 denote the optimal levels of utility before and 
after the price increase.
> umax1:=evalf(umx1);

> umax2:=evalf(umx2);

 Here are the x and y intercepts for the two budget constraints. 
> x1int:=round(M/p1[x]);

> x2int:=round(M/p2[x]);

> yint:=M/p[y];

 Arange in functional form for the purpose of graphing the budget constraints. 
> BC1:=x->yint-(p1[x]/p[y])*x;

> BC2:=x->yint-(p2[x]/p[y])*x;

 Solve and graph the two indifference curves indicating utility before and after the 
price increase. 
> indy1:=solve(u=umax1,y);

> indy2:=solve(u=umax2,y);

> indcv1:=x->indy1;

> indcv2:=x->indy2;

 The graph below illustrates the optimal solutions before (red and yellow tangency) 
and after (green and blue tangency) the price of x increases.
> 
plot([BC1(x),BC2(x),indcv1(x),indcv2(x)],x=0..x1int,y=0..yi
nt,labels=["good x","good y"],title="Figure 5: Consumer 
Equilibrium",thickness=4,axes=boxed);
 

 

 
 To identify the Hicks substitution and income effects for x, look at the choice the 
consumer would make if allowed to maintain original utility level at the new price 
ratio.
> pr:=p2[x]/p[y];

> u1:=umax1-u;

> u1y:=solve(u1=0,y);

 Find the marginal rate of substitution. 
> Diff(u1y,x);

> u1slope:=-value(%);

 Use the fact that at the point of tangency, the price ratio equals the marginal rate of 
substitution in order to find the x and y values associated with the Hicksian solution. 
> Hicks:=solve(pr=u1slope,x);

> Hicksx:=max(Hicks);

> Hicksy:=eval(u1y,x=Hicksx);

 Numerically solve for the Hicks substitution and income effects. 
> x1:=solve(soln1[3]); 

> SubEffect:=x1-Hicksx; 

> x2:=solve(soln2[3]); 

> IncEffect:=Hicksx-x2; 

 Next, rearrange format for graphing the Hicksian budget constraint. 
> Mg:=p2[x]*Hicksx+p[y]*Hicksy;

> BC3:=x->(Mg/p[y])-(p2[x]/p[y])*x;

 The graph below indicates the Hicks substitution and income effects
> 
plot([BC1(x),BC2(x),indcv1(x),indcv2(x),BC3(x)],x=0..x1int,
y=0..yint,labels=["good x","good y"],title="Figure 6: Hicks 

Substitution and Income Effects for Good 
x",thickness=4,axes=boxed); 
 

 
 
 Questions:
 On the graph,
 1)  identify the original optimal bundle,
 2)  identify the optimal bundle after the price change,
 3)  identify the change in good x that describes the substitution effect,
 4)  identify the change in good x that describes the income effect.
 5)  Repeat (3) and (4) for good y.
 
 

The next four examples come from intermediate macroeconomic theory.  The first 
example depicts the Keynesian Cross.  The second solves the Keynesian IS-LM model.  
The final two examples illustrate the Solow model of economic growth, including the 
Golden Rule. 

Example 5: 
 
The Keynesian Cross  
> restart;
> assume(Y1>=0,TAX>=0,INV>=0,GOV>=0,IM>=0);
This program allows us to look at nonlinear versions of the 
model. 
> 
Government Expenditures and the Marginal Tax Rate 
> 
> GOV:=1000-0.01*Y;

> TAX:=0.05*Y;

> 
The Consumption Function 
> 
> CONS:=3000+15*(Y-TAX)^0.5;

The Investment Schedule 
> INV:=1000+0.1*Y;

> 
Exogenous Exports 
> EX:=600;

> 
Imports 
> 
> IM:=400+0.1*(Y-TAX);

Define the Keynesian Cross 
> AGGEXP:=CONS+GOV+INV+EX-IM;

> 
Equilibrium GDP for this Model 
> GDP:=solve({AGGEXP=Y});

Define Savings 
> SAV:=INV+GOV+EX-IM-TAX;

Define Disposable Income 
> DISPINC:=CONS+SAV;

Evaluate these parameters with respect to equilibrium GDP. 
> eval([Y,SAV,DISPINC],GDP);

> eval([TAX,INV,GOV,IM,EX],GDP); 

> 
p1:=plot([AGGEXP],Y=6000..7000,labels=["GDP","Ex"],title="F
igure 7, Aggregate Expenditures",thickness=4,color=blue): 
> f:=Y->Y; 

> p2:=plot([f(Y)],Y=6000..7000,thickness=2):
> with(plots): 
Warning, the name changecoords has been redefined 
 
> display(p1,p2); 

 
 
 
 
 
 
 
 
 
Example 6:

The Keynesian IS-LM Model
  
> restart; 

 
Government Expenditures and the Marginal Tax Rate
> GOV:=10000; 

> TAX:=0.05*Y; 

The Consumption Function
> CONS:=8000+.85*(Y-TAX); 

The Investment Schedule
> INV:=2000-50*r; 

Exports
> EX:=6000; 

Imports
> IM:=5000+0.05*(Y-TAX); 

Determine the Investment-Savings (IS) Function.
> AGGEXP:=CONS+GOV+INV+EX-IM; 

> IS:=solve(Y=AGGEXP,r); 

> p1:=plot(IS, Y=85000..90000,r=0..15,thickness=5): 
>  
>  
Given the Money Supply. 
> MS:=8500; 

Given Money Demand. 
> MD:=40+0.1*Y-30*r; 

Solve for Liquidity-Money (LM) Curve. 
> LM:=solve(MS=MD,r); 

>  
> p2:=plot(LM, Y=85000..90000,r=0..15,title="Figure 8: The 
Keyesian IS-LM Curves",thickness=4,color=blue): 
> with(plots): 
 
> display(p1,p2,labels=["Y","r"]); 

Solve for the Equilibrium GDP and Interest Rate. 
Equilibrium GDP: 
> eY:=solve(IS=LM,{Y}); 

Equilibrium Interest Rate: 
> er:=eval(LM,eY); 

Example 7: 
   
The Solow Growth Model 
 
Though not essential for this application, the purpose of the inactive code (in black) 
is to provide related information and to more clearly reveal how the model is 
constructed. 
 
> restart; 
 
1)  Define a CRS Production Function. 
F:=(K,L)->A*K^.5*L^.5; 
Diff(F(K,L),L); 
MPL:=value(%); 
Diff(F(K,L),K); 
MPK:=value(%); 
> A:=3; 

k:=K/L; 
y:=Y/L; 
Redefine Production in terms of Relative Capital. 
> f:=k->A*k^.5; 

2)  Incorporate Investment, Consumption and Savings. 
Y:=C+S; 
C:=(1-sigma)*Y; 
c:=C/L; 
s:=S/L; 
i:=I/L; 
Leakages = Injections in Equilibrium. 
s:=i; 
Let sigma represent the marginal propensity to save.  Then, 1-sigma is the marginal 
propensity to consume. 
c:=(1-sigma)*y; 
c:=y-s; 
i=sigma*y; 
> sigma:=.15; 

 3)  The Capital Constraint 
ChangeK:=I-delta*k; 
> delta:=.10; 

> changek:= sigma*f(k)-delta*k; 

changek is the change in effective (K/L) capital. 
 
4)  Consumption 
> cc:=f(k)-changek-delta*k; 

cc is consumption. 
> p1:=plot(f(k),k=1..1000,title="Figure 9:  Production and 
Consumption",color=blue,thickness=3): 
> p2:=plot(delta*k,k=1..1000, thickness=4): 
> p3:=plot(cc,k=1..1000, color=brown,thickness=5): 
 
Production, f(k), is in blue.  Consumption is in brown.  Depreciation is in red. 
> with(plots): 
 
> display(p1,p2,p3); 

 
  
 
 The final example is an extension of example 7.  It 
depicts the Golden Rule for capital accumulation.  We 
derive the steady state level of capital necessary to 
maximize per capita consumption over time.   

Example 8:
 
The Golden Rule:  Determining Optimal Steady State 
Consumption from the Solow Growth Model 
  
> restart; 
> A:=4; 

> f:=k->A*k^.5; 

>  
> delta:=.10; 

> Diff(f(k),k); 

> MPk:=value(%); 

> grule:=solve(MPk=delta,k); 

The Golden Rule:  grule is the optimal steady state level of capital associated with the 
highest possible sustainable level of consumption.  It is given by the intersection of the 
savings (green) and depreciation (red) curves below.  
> sigma:=delta*grule/f(grule); 

>  Sigma is the optimal savings rate that guarantees a 
maximum steady state stream of consumption. 
> p1:=plot(f(k),k=1..1000,title="Figure 10:  Golden Rule 
Production",color=blue,thickness=3): 
> p2:=plot(delta*k,k=1..1000,thickness=4): 
> p3:=plot(sigma*f(k),k=1..1000,color=green,thickness=5): 
> with(plots): 
 
>  Production, f(k), is in blue.  Depreciation is red.  
Savings is green. 
> display(p1,p2,p3); 

